Clinical Trial Simulation Using Wings for NONMEM

Objective:

1. To provide practical experience of performing a clinical trial simulation.
2. To show how to use Wings for NONMEM and NONMEM to simulate a clinical trial
3. To demonstrate how to analyze clinical trial simulation results using Excel

Introduction

A data set is provided which comes from a randomized concentration-controlled trial of theophylline.

Holford N, Black P, Couch R, Kennedy J, Briant R. Theophylline target concentration in severe airways obstruction - 10 or 20 mg/L? A randomised concentration-controlled trial. Clin Pharmacokinet. 1993a;25(6):495-505.

Figure 1 Time course of peak expiratory flow rate in patients randomized to target theophylline concentrations of 10 mg/l or 20 mg/L (Holford et al. 1993a)



Fig. 1. Time course of theophylline concentrations. Low = 10 mg/L target concentration group; High = 20 mg/L target concentration group; Pre = before theophylline administration in the trial; 1, 5, 24, 48 = 1, 5, 24 and 48 hours after entry to the trial, respectively; End IV = at end of intravenous theophylline infusion; DC = at the time of discharge from hospital ward; OP = at the outpatient clinic.

A model based analysis of this data (Holford et al. 1993b) showed that both time and theophylline concentration were important covariates for description of the peak expiratory flow rate (PEFR) response to treatment.

Holford N, Hashimoto Y, Sheiner LB. Time and theophylline concentration help explain the recovery of peak flow following acute airways obstruction. Population analysis of a randomised concentration controlled trial. Clin Pharmacokinet. 1993b;25(6):506-15.

Hands OnSteps

NM-TRAN control streams will be used to illustrate how to simulate this clinical trial with a simplified model using only theophylline concentration as the explanatory variable.

1. Open the WFN_CTS shortcut which will start WFN in the CTS\NM72 folder



2. Use *nmgo* to run the *theopd_org* problem.

```
C:\Docs\Mtg\12\PAGANZ Monash\PAWS\Intermediate\CTS\NM72>nmgo theopd_org
C:\nm720\ver7_20110527\gf\reg\tr\nmtran.exe theopd_org.ctl

 WARNINGS AND ERRORS (IF ANY) FOR PROBLEM    1

 (WARNING  2) NM-TRAN INFERS THAT THE DATA ARE POPULATION.

 (WARNING  13) WITH USER-WRITTEN PRED OR $PRED, NM-TRAN CANNOT APPEND THE
 MDV DATA ITEM.
 CREATING MUMODEL ROUTINE...
Completed call to gfcompile.bat
NONMEM theopd_org...
```

This illustrates the use of an Emax model to describe the PEFR changes observed in the original concentration-controlled trial

```
THETA:       POP_S0       POP_EMAX     POP_C50      RUV_SD
ETA:         PPV_S0       PPV_EMAX     PPV_C50
ERR:         EPS1
theopd_org.lst   5800.756          FOCE     eval=381 sig=3.3 sub=153 obs=574 MOD(NONMEM)7.2.0
THETA      = 151          167          8.49         79.9

ETASD      = 0.408656     0.556776     1.10905
ETAR12     =-0.185
ETAR23     = 0.960 -0.335

ETAPval    = 0.012979     0.80111      0.010291
ETAshr%    = 57.918       25.185       58.299
EPSshr%    = 10.556

EPSSD      = 1

MINIMIZATION SUCCESSFUL
HOWEVER, PROBLEMS OCCURRED WITH THE MINIMIZATION.
REGARD THE RESULTS OF THE ESTIMATION STEP CAREFULLY, AND ACCEPT THEM ONLY
AFTER CHECKING THAT THE COVARIANCE STEP PRODUCES REASONABLE OUTPUT.
```

3. Open the *theopd_org.ctl* file and save it as *theopd_sim.ctl*

4. Edit *theopd_sim.ctl* to simulate PEFR

```
$PROB theophylline concentration controlled trial
$INPUT
ID ; patient ID
TIME ; hours since entry to trial
CONC ; theophylline concentration mg/L
DV ; peak expiratory flow rate L/min

;Data from Holford N, Hashimoto Y, Sheiner LB. Time and theophylline
;concentration help explain the recovery of peak flow following acute
;airways obstruction. Population analysis of a randomised concentration
;controlled trial. Clin Pharmacokinet. 1993b;25(6):506-15
$DATA theopd_org.csv

$SIM (20120205) (20120206 UNIFORM) ONLYSIM NSUB=1
$THETA 0.5 FIX  ; F_10 fraction randomized to 10 mg/L target conc
$OMEGA 0.25 FIX ; PPV_CONC 50% BSV in concs

$THETA
;Pharmacodynamics
(0,150.,)    ; POP_S0 L/min
(0,200.,)    ; POP_EMAX L/min
(.1,10,20)     ; POP_C50 mg/L
;Residual error
(0,10,)      ; RUV_SD L/min

$OMEGA BLOCK(3)
0.1 ; PPV_S0
0.01 0.1 ; PPV_EMAX
0.01 0.01 0.1 ; PPV_C50

$SIGMA 1 FIX       ; EPS1

$PRED
IF (ICALL.EQ.4) THEN ; simulation
   IF (NEWIND.LE.1) THEN ; first record of each subject
      CALL RANDOM(2,R)
      ;randomize to 10 mg/L or 20 mg/L target
      IF (R.LE.F_10) THEN
         TARGET=10
      ELSE
         TARGET=20
      ENDIF
   ENDIF
   ; every record
   TRT=TARGET ; simulated treatment group
   CONC=TARGET*EXP(PPV_CONC) ; add BSV to simulated conc
ENDIF
S0=POP_S0*EXP(PPV_S0)
EMAX=POP_EMAX*(1+PPV_EMAX)  ; note proportional model for Emax
C50=POP_C50*EXP(PPV_C50)
Y = S0 + EMAX*CONC/(CONC+C50) + RUV_SD*EPS1

$TABLE ID TRT TIME CONC DV
NOAPPEND ONEHEADER NOPRINT FILE=sim.fit
```

5. Run *theopd_sim* to simulate treatment groups, concentrations and PEFR values. Open the *theopd_sim.reg/ theopd_sim.fit* file with Excel, delete row 1, select all (ctrl-A) and copy (ctrl-C) then paste cols A-E of *theopd_sim.fit.xlsx*.starting at cell A1 (ctrl-V). Format cells General so they are more readable (Figure 2).

Figure 2 Simulated treatments, concentrations and PEFR using theopd_sim

	ID	TRT	TIME	CONC	DV	MDV
2	1	20	0.01	17.828	189.86	0
3	1	20	1.51	17.828	194.21	0
4	2	20	0.01	22.045	277.05	0
5	2	20	1.42	22.045	280.16	0
6	2	20	4.42	22.045	289.95	0
7	3	20	0.01	21.702	162.05	0
8	3	20	1.01	21.702	158.08	0
9	3	20	3.76	21.702	171.42	0
10	4	10	0.01	6.3238	131.16	0
11	4	10	1.09	6.3238	174.08	0
12	4	10	9.59	6.3238	166.07	0
13	4	10	18.34	6.3238	151.71	0
14	5	10	0.01	5.5932	290.87	0
15	5	10	1.26	5.5932	311.8	0
16	7	10	0.01	14.3	217.22	0
17	8	20	0.01	31.724	211.48	0
18	8	20	1.34	31.724	210.21	0
19	8	20	5.26	31.724	200.65	0
20	9	10	0.01	5.2289	364.49	0
21	9	10	1.76	5.2289	376.07	0
22	11	10	0.01	15.848	397.89	0
23	11	10	1.51	15.848	387.61	0
24	11	10	4.01	15.848	380.69	0
25	12	20	1.01	7.4099	119.22	0
26	12	20	9.01	7.4099	124.54	0
27	12	20	302.26	7.4099	98.284	0
28	13	20	0.01	36.562	216.99	0
29	13	20	1.01	36.562	204.56	0
30	13	20	13.01	36.562	200.95	0
31	13	20	37.01	36.562	198.32	0
32	13	20	43.76	36.562	220.35	0
33	13	20	60.76	36.562	197.44	0
34	13	20	376.01	36.562	207.78	0





Note that there are some negative PEFR values because of the additive residual error which uses a normal distribution.

6. The simulation code may be modified to avoid this problem. The NEW option is required for the first random number generator when using the NONMEM built-in function SIMEPS().

```
$SIM (20120205 NEW) (20120206 UNIFORM) ONLYSIM NSUB=1

Y = S0 + EMAX*CONC/(CONC+C50) + RUV_SD*EPS1
IF (ICALL.EQ.4) THEN
   NEPS=0
   DOWHILE (Y.LE.50.AND.NEPS.LT.100)
      CALL SIMEPS(EPS)
      Y = S0 + EMAX*CONC/(CONC+C50) + RUV_SD*EPS1
      NEPS=NEPS+1
   ENDDO
   IF (NEPS.EQ.100) THEN
      MDVX=1
   ELSE
      MDVX=0
   ENDIF
ENDIF

$TABLE ID TRT TIME CONC DV MDVX
```

7. Edit *theopd_sim.ctl* to simulate truncated PEFR distribution and save it as *theopd_sim_trunc.ctl*. Then run *theopd_sim_trunc*. Repeat the instructions in Step 5 to look at the results in Excel. The PEFR distribution has been truncated with a lower bound of 50 L/min (Figure 3).

Figure 3 Simulated treatments, concentrations and truncated PEFR using theopd_sim_trunc



8. The simulated data can then be used to evaluate the power of the design using different models relating treatment group or concentration to PEFR. A model for the null hypothesis that there is no difference between the treatments (*trial_placebo_est.ctl*) is used for comparison with a model using treatment group (*trial_trt_est.ctl*) or concentration (*trial_conc_est.ctl*). Look at these control stream files to see how these different models are constructed.

9. The WFN command *nmgosim* can be used to simulate multiple instances of the trial and estimate parameters and objective function values using the different models. Here is the content of *sim.bat*. This is aWindows command batch file which calls *nmgosim* for each of the models.

Change the simulation record seed number in your copy of *theopd_sim_trunc.ctl* so that your simulations will be different.

```
rem To create and delete simulated data: set ctsthisgotdata=
rem To create and keep simulated data:   set ctsthisgotdata=n
rem To skip creation and keep sim data:  set ctsthisgotdata=y
rem Non-default simulated data dir:      set ctsdata=non_default_dir

rem create and keep simulated data, estimate with placebo model
set ctsthisgotdata=n
call nmgosim theopd_sim_trunc trial_placebo_est 1 10

rem use simulated data, estimate with treatment model
set ctsthisgotdata=y
call nmgosim theopd_sim_trunc trial_trt_est 1 10

rem use simulated data, estimate with concentration model
set ctsthisgotdata=y
call nmgosim theopd_sim_trunc trial_conc_est 1 10
```

10. After running *sim.bat* open the **.smy* file in the run folder for each model.

11. Open the Excel file *nmgosim.xlsx* and paste the contents of each *smy* file into the appropriate worksheet.

12. Look at the *Bias&Imprecision* and the *PowerTrt* and *PowerConc* worksheets to evaluate the clinical trial simulation

13. How do you think the clinical trial design could be changed in order to increase the power of showing a treatment effect?